

Tutorial category: Normal mode

# First analysis with MadAnalysis 5



Version 1.0

Date 23/07/2016

Official MadAnalysis 5 website : <https://launchpad.net/madanalysis5/>

# Goals of this tutorial

- Defining an elementary analysis in MadAnalysis 5 including plots and cut selection.
- Launching the analysis over the datasets and getting the results.
- To be able to use the Python console or a script for running an analysis.
- Changing the way to normalize or to stack the histograms.
- Selecting the figure of merit in a selection.

# Requirements

- MadAnalysis 5 is installed on your system and has been launched successfully at least one time. The collection of example samples is installed too.



Part 1

# A basic example

# A basic example

Type this example at the prompt of MadAnalysis 5.

```
ma5> define mu = mu+ mu-  
ma5> import samples/ttbar_*.lhe.gz  
ma5> plot PT(mu)  
ma5> reject PT(mu) < 40  
ma5> plot PT(mu)  
ma5> submit  
ma5> open
```

# A basic example

```
ma5> define mu = mu+ mu-
```

Create a shortcut label 'mu' for considering both muons and antimuons.

```
ma5> import samples/ttbar_*.lhe.gz
```

Import all the samples beginning with *ttbar\_* and finishing with *.lhe.gz*.

```
ma5> plot PT(mu)
```

Add to the analysis a plot of (anti)muons transverse momentum.

```
ma5> reject PT(mu) < 40
```

Add to the analysis a cut removing all muons with a transverse momentum less than 20 GeV.

```
ma5> plot PT(mu)
```

Add (again) to the analysis a plot of (anti)muons transverse momentum.

```
ma5> submit
```

Launch the analysis over the samples.

```
ma5> open
```

Open the report containing the results of the analysis.

# A basic example in script format

Note that this analysis could have been written in a text file ...

## MyFirstExample.txt

```
define mu = mu+ mu-  
import samples/ttbar_*.lhe.gz  
plot PT(mu)  
reject PT(mu) < 40  
plot PT(mu)  
submit  
open
```

... and executed by the following command:

```
./bin/ma5 MyFirstExample.txt
```



Part 2

# Building your analysis step-by-step

# Analysis definition in 5 steps

In this section, we describe the different steps building a classical analysis with MadAnalysis 5 normal mode:

1. Defining your proper particle labels
2. Importing your samples
3. Defining your analysis by adding:
  - 3.1 some plots
  - 3.2 some cuts
4. Launching the analysis
5. Opening the report

# 1. Defining (multi)particle labels

- Particles are defined by labels which point to one PDG-id.
- A « multiparticle » correspond to a label pointing to several PDG-ids.
- SM and MSSM labels are automatically loaded at the starting of MadAnalysis.
- The command `display_particles` & `display_multiparticles` allow you to display all the labels predefined.

```
ma5>display_particles
```

```
MA5: a b b1 b1~ b2 b2~ b~ c cl cl~ cr cr~ c~ d dl dl~ dr dr~ d~ e+ e- el+
el- er+ er- g go h+ h- h1 h2 h3 met mht mu+ mu- mul+ mul- mur+ mur- n1 n2
n3 n4 s sl sl~ sr sr~ sve sve~ svm svm~ svt svt~ s~ t t1 t1~ t2 t2~ ta+ ta-
ta1+ ta1- ta2+ ta2- t~ u ul ul~ ur ur~ u~ ve ve~ vm vm~ vt vt~ w+ w- x1+
x1- x2+ x2- z
```

```
ma5>display_multiparticles
```

```
MA5: hadronic invisible j l+ l- p vl vl~
```

- There are 2 special labels :
  - `hadronic`: contains all hadronic particles
  - `invisible`: contains all invisible particles

# 1. Defining (multi)particle labels

- To show to which PDG-id code a label refers:

```
ma5>display n1  
The particle 'n1' is defined by the PDG-id 1000022.
```

- To create a new label for a PDG-id (useful for BSM particle):

```
ma5>define extrapart = 5000001
```

- To create a new label gathering several particles (useful in the analysis definition):

```
ma5>define extrapart = 5000001 -5000001  
ma5>define mu = mu+ mu-  
ma5>define b = b b~
```

## 2. Importing your samples

- For MadAnalysis, a dataset is a collection of samples which will be merged.
- All sample files are stored in a dataset.

```
ma5> import samples/tt*.lhe
```



A default dataset called `defaultset` is created

```
ma5> import samples/ttbar_sl_1.lhe.gz as ttbar
```

```
ma5> import samples/Wj*.lhe as Wjets
```

```
ma5> import samples/ttbar_sl_2.lhe.gz as ttbar
```



Two datasets are created: `ttbar` and `Wjets`

- Possible to display all the datasets defined by the user with the command `display_datasets`.

```
ma5>display_datasets
***** List of defined datasets *****
defaultset (signal)
Wjets      (signal)
ttbar      (signal)
```

## 2. Importing your samples

Properties of a dataset could be changed by the command `set`.

Some relevant properties:

- Possibility to tag datasets as signal or background (useful for selection and computation of the figure of merit).

```
ma5> set ttbar.type = signal  
ma5> set Wjets.type = background
```

- Possibility to force the cross section value (normally this value is extracted from the sample). Unit = pb.

```
ma5> set ttbar.xsection = 1.0
```

- Possibility to multiply the cross section by a weight. For instance, this weight could be k-factor.

```
ma5> set defaultset.weight = 1.2
```

## 3.1. Plots

- Observable can be related to the event or the properties of a particle
- Plethora of observables: N, E, ET, M, MT, P, PT, PX, PY, PZ, THETA, ETA, ...,

```
ma5> plot NPID
ma5> plot MET
ma5> plot PT(mu)
```

- Combining particles is possible (vector sum of 4-vector momenta)

```
ma5> plot M(mu+ mu-)
```

- Number of bins and x-axis bounds are set by default. This value can be specified by the user if necessary.

```
ma5> plot M(mu+ mu-) 100 50 150
```

number of bins    xmin    xmax

## 3.1. Plots

- **List of observables related to the events:**
  - multiplicity (`N`, `NPID`, `NPID`),
  - Total energy (`TET`) or total hadronic energy (`THT`),
  - Missing transverse energy (`MET`) or missing transverse hadronic energy (`MHT`)
  - SUSY transverse observables (`ALPHAT`, `MT2`)
  - Coupling/scales (`ALPHA_QCD`, `ALPHA_QED`, `SQRTS`, `SCALE`)
- **List of observables related to one particle:**
  - energy (`E` and `ET`),
  - mass (`M`, `MT`, `MT_MET`),
  - momentum magnitude and components (`P`, `PT`, `PX`, `PY`, `PZ`),
  - angles (`THETA`, `ETA`, `ABSETA` and `PHI`),
  - relativist factors (`Y`, `BETA` and `GAMMA`).
- **List of observables related to tow particles:**
  - Manhattan distance in eta-phi plane (`DELTAR`),



## 3.1. Plots

Several options or syntaxes allow to extend the potential of MadAnalysis. 3 examples:

- By default, a combination is interpreted as the vector sum of momenta. This interpretation can be changed by adding a prefix to the observable label.

For instance : `vPT`, `sPT`, `dsPT`, `dvPT`, `rPT`

- Selecting a particle according to its rank in energy (or to other observables)

```
ma5> plot PT(mu+[1])
```

- Selecting a particle according to its history (requirements on mother, grand-mother ...)

```
ma5> plot PT(mu+ < w+ < t~)
```

# 3.1. Plots

- Options can be added to plots into square brackets [ ]

```
ma5> plot M(mu+ mu-) [ logX ]
```



Log scale for x-axis

```
ma5> plot M(mu+ mu-) [ logY ]
```



Log scale for y-axis

```
ma5> plot M(mu+ mu-) [ logX logY ]
```



Log scale for both axes

```
ma5> plot M(mu+ mu-) [ finalstate ]
```



Only particles in the final state (default)

```
ma5> plot M(mu+ mu-) [ initialstate ]
```



Only particles in the initial state

```
ma5> plot M(mu+ mu-) [ innerstate ]
```



Only particles which have been produced and have decayed.

## 3.2. Selection cuts

- Selection cuts can be performed by two commands: select and reject. These two commands are equivalent.
- Cuts : selecting / rejecting events

```
ma5> reject MHT < 50  
ma5> select N(mu) >= 2
```

- Cuts : selecting / rejecting a particle or a combination

```
ma5> select (mu) PT > 50  
ma5> select 80 < M (mu+ mu-) < 100
```

- The list of observables and syntax are identical to histograms ones.

## 4. Launching the analysis over the samples

This can be done by the command `submit`

- Creating a working directory (with a default name if no name is specified)
- Compiling the C++ job
- Launching the analysis over the different samples contained in the datasets


```
ma5> submit
[...]
```

```
* SampleAnalyzer 2.0 for MadAnalysis 5 - Welcome.
* Option choices: selecting analysis = 'MadAnalysis5job'.
* Extracting the following sample files:
* 1/4 ~/samples/ttbar_sl_1.lhe.gz
=> file size : 107.09 Mo
=> sample produced by MadGraph.
=> progress [=====> ]
```

If you modify, after the submission, the analysis or the layout of the plots , the results can be updated in an optimized way by the command `resubmit`.

# 5. Opening the analysis report

The command `open` displays the HTML report of the last job created. Reports in PDF and DVI format are also available.

 Please visit us.	<h2 style="margin: 0;">MadAnalysis 5 report</h2> <p style="font-size: small; color: blue;">Created by <i>econte</i> on 05 November 2012, 21:29:45</p>
---	---

**PDF version of this report**

- [Download here](#)

---

**Setup**

---

- [Command history](#)
- [Configuration](#)

---

**Datasets**

---

- [defaultset](#)

---

**Histos and cuts**

---

- [Histogram 1](#)
- [Histogram 2](#)
- [Cut 1](#)
- [Cut 2](#)
- [Histogram 3](#)

---

**Summary**

---

- [Cut-flow chart](#)

**Setup**

---

**Command history**

```

ma5>define mu = mu+ mu-
ma5>import samples/ttbar_sl_1.lhe.gz
ma5>import samples/ttbar_sl_2.lhe.gz
ma5>import samples/ttbar_fh.lhe.gz
ma5>import samples/zz.lhe.gz
ma5>ma5>plot MET
ma5>ma5>plot PT(mu) 20 0 100
ma5>ma5>reject MET > 100
ma5>ma5>reject (mu) PT < 20
ma5>ma5>plot M(mu+ mu-) 20 0 100
ma5>ma5>submit
ma5>plot MET
ma5>plot PT(mu) 20 0 100
ma5>reject MET > 100
ma5>reject (mu) PT < 20
ma5>plot M(mu+ mu-) 20 0 100
ma5>submit
          
```

**Configuration**

Part 3

# Some useful main options

# Options related to histogramming

- The histograms are by default normalized to the integrated luminosity. Integrated luminosity is by default  $10 \text{ fb}^{-1}$ . This value can be set by the user:

```
ma5> set main.lumi = 100
```

- The histograms can be normalized to one by the following command:

```
ma5> set main.stacking_method = normalize2one
```

- If you have several datasets, you can choose if you would like to stack the spectra (by default) or to superimpose them.

```
ma5> set main.stacking_method = stack
```

```
ma5> set main.stacking_method = superimpose
```

# Figure of merit for the selection

- If you apply on datasets (with signal and background tags), MadAnalysis 5 will compute a cut-flow chart including a figure of merit.
- He must choose one formula among a predefined collection (by default option 4 is used).

- 1:  $S/B$
- 2:  $S/\sqrt{B}$
- 3:  $S/(S+B)$
- 4:  $S/\sqrt{S+B}$
- 5:  $S/\sqrt{S+B+(xB)**2}$

with S and B mean respectively  
Signal and Background

- Corresponding instruction in MadAnalysis 5.

```
ma5>set main.fom.formula = <formula number>
```

- Special case: the x parameter in formula number 5

```
ma5>set main.fom.formula = 5
ma5>set main.fom.x = 0.2
```





# About this document

- The present document is a part of the tutorial collection of the package MadAnalysis 5 (MA5 in abbreviated form). It has to be conceived to explain in a practical and graphical way the functionalities and the various options available in the last public release of MA5.
- The up-to-date version of this document, also the complete collection of tutorials, can be found on the MadAnalysis 5 website :

<https://madanalysis.irmp.ucl.ac.be/wiki/tutorials>

- Your feedback interests ourselves (bug reports, questions, comments, suggestions). You can contact the MadAnalysis 5 team by the email address : [ma5team@iphc.cnrs.fr](mailto:ma5team@iphc.cnrs.fr)

